Parametric Curves and Quadratic Surfaces with Maple[1]

Preliminaries

Start Maple from the `Start` button `-> Programs -> Math -> Maple 11.0 -> Maple 11`. Go to the **Tools** menu and select **Options**. A dialog window will pop up, click on **Display** and select **window** for **Plot display**, then click the **Apply to Session** button at the bottom of the box (this will make each plot appear on a separate window).

At the prompt, type:

```
> with(plots):
```

This will allow you to use different plotting commands available in Maple.

Don't forget to type a semicolon (;) (or a colon (:) if you don't want to see the output) at the end of each command!!!

Plotting Parametric Curves

The `plot` command in Maple allows you to plot parametric curves using the syntax:

```
> plot([x(t),y(t),t=t_0..t_f], options);
```

This will produce the parametric curve defined by `(x(t),y(t))` when the parameter `t` ranges from `t_0` to `t_f`. Different *options* are available for color, axis, scaling, etc. Try this:

```
> plot([cos(t), sin(t), t=0..2*Pi], scaling=constrained);
```

Now, let's define some functions of `t` and produce different parametric plots with them:

```
> f1 := t-> 3*cos(2*t) ; g1 := t-> sin(4*t) ;
> plot([f1(t), g1(t), t=0..2*Pi]);
> f2 := t-> cos(t)/ln(t) ; g2 := t-> sin(t)/ln(t) ;
> plot([f2(t), g2(t), t=3..20]);
> plot({[f1(t), g1(t), t=0..2*Pi], [f2(t), g2(t), t=3..20]});
```

Try different combinations of `f1(t)`, `f2(t)`, `g1(t)`, and `g2(t)` over different intervals of `t`. You can also create animations to visualize (and understand) the effects that different scaling factors have in a given curve; try, for example:

```
> animate([2*cos(3*t), b*sin(2*t), t=0..2*Pi, numpoints=500], b=2..10);
```

and

[1] This worksheet has been prepared with the help of online materials available from Maple's Application Center.

```
> animate([2*cos(3*t), (6 + 4*sin(b))*sin(2*t), t=0..2*Pi], b=0..2*Pi);
```

Play with the controls in the plot window and see if there are any differences between the two animations.

Plotting Parametric Curves in 3d

The Maple command `spacecurve` allows you to plot parametric curves in space. Try, for instance, the parametric equations of a line:

```
> spacecurve([1+ t, 2 - 3*t, -1 - t, t=0..10]);
```

Here are some other examples:

```
> spacecurve([cos(5*t), sin(5*t), t, t=0..Pi]);
> spacecurve([cos(5*t), sin(5*t), log(t), t=0..Pi]);
> spacecurve([t, 1/(1+t^2), t^2, t=-10..10]);
> spacecurve([cos(t), sin(t), sin(5*t), t=0..2*Pi]);
```

Plotting Cylinders and Quadratic Surfaces

To plot cylinders and quadratic surfaces in Maple we can use the `implicitplot3d` command. The syntax for this command is (no need to type this):

> `implicitplot3d(` *expression*, `x=a..b, y=c..d, z=e..f,` *options*`);`

where *expression* is some expression involving `x, y,` and `z`, and `a, b, c, d, e,` and `f` real constants specifying the ranges of `x, y,` and `z`. For example, a cylinder can be plotted using:

```
> implicitplot3d(x^2 + y^2 = 1, x=-1..1, y=-1..1, z=-3..3);
> implicitplot3d(x^2 + z^2 = 1, x=-1..1, y=-3..3, z=-1..1);
```

Try now some of the quadratic surfaces described in the table at the end of section 11.8 (pg. 605 – 606). There are three types you should get familiar with and be able to recognize and sketch (roughly):

Ellipsoids:

```
> implicitplot3d((1/4)*x^2 + (1/9)*y^2 + (1/16)*z^2 = 1, x=-2..2, y=-3..3, z=-4..4);
```

Paraboloids:

```
> implicitplot3d(z=(1/4)*x^2 + (1/9)*y^2, x=-4..4, y=-6..6, z=0..6, scaling=constrained);
> implicitplot3d(y=(1/4)*x^2 + (1/9)*z^2, x=-4..4, y=0..6, z=-6..6, scaling=constrained);
> implicitplot3d(x=4*(y-2)^2 + (z-2)^2, x=0..10, y=-1..5, z=-2..6);
```

Cones:

```
> implicitplot3d(z^2=(1/4)*x^2 + (1/9)*y^2, x=-3..3, y=-4..4, z=-1..1);
> implicitplot3d(y^2=(1/4)*x^2 + (1/9)*z^2, x=-3..3, y=-1..1, z=-4..4);
```

Other quadratic surfaces that we'll come across are hyperboloids:

Of One Sheet:

```
> implicitplot3d((1/4)*x^2 + (1/9)*y^2 - z^2 =1, x=-5..5, y=-7..7, z=-2..2);
> implicitplot3d(-x^2 + (1/9)*y^2 + (1/4)*z^2 =1, x=-2..2, y=-7..7, z=-5..5);
```

Of Two Sheets:

```
> implicitplot3d((1/4)*x^2 - (1/9)*y^2 - z^2 =1, x=-4..4, y=-6..6, z=-2..2);
```

And **Hyperbolic Paraboloids:**

```
> implicitplot3d(z = - (1/4)*x^2 + (1/9)*y^2, x=-5..5, y=-6..6, z=-2..2);
```

Plotting in Cylindrical and Spherical Coordinates

Maple can also plot surfaces expressed in cylindrical and spherical coordinates using the `plot3d` command:

```
> plot3d(r(theta,z), theta=a..b, z=c..d, coords=cylindrical);
> plot3d(rho(theta,phi), theta=a..b, phi=c..d, coords=spherical);
```

Here are some examples:

Cylindrical Coordinates:

```
> plot3d(1, theta=0..2*Pi, z=-3..3, coords=cylindrical);
> plot3d(z, theta=0..2*Pi, z=0..3, coords=cylindrical);
```

Spherical Coordinates:

```
> plot3d(1, theta=0..2*Pi, phi=0..Pi, coords=spherical);
> plot3d(sin(theta)*cos(phi), theta=0..2*Pi, phi=0..Pi, coords=spherical);
```

That's all! Save this hand out as it may be useful for the computer project that will be assigned with homework #5.